//Подключаем инициализацию дополнительных переменных
#include "scripts\bgr_vars.sqf"
//Получаем значения параметров из дескрипшена
#include "scripts\processParamsArray.sqf";
//Завершение миссии через команду: "Победа!" call f_bgrEnd;
#include "scripts\bgr_endMission.sqf";

//Прячем маркера
0 = [["zoneB","startB"],resistance] execVM "scripts\bgr_HideMarkers.sqf";
0 = [["zoneR","startR"],west] execVM "scripts\bgr_HideMarkers.sqf";

//Выключаем сохранения
enableSaving [false, false];

//Disable BIS Conversations
player setVariable ["BIS_noCoreConversations", true];

//Фризтайм
0 = [brifingTime] call bgr_fnc_frizTime;
//0 = [brifingTime,["zone_marker1","zone_marker2"],50] call bgr_fnc_frizTime; //время из дескрипшена, 50 метров если не в зоне маркера

//Брифинг
_briefscr = execVM "briefings\init.sqf";

//Боты за которых играют не двигаются
{_x disableAI "ANIM";} foreach bgr_PlayableUnits;

//Сообщение при старте миссии
["Stratis", "2035", "www.our-army.su"] spawn { sleep 0.1; waitUntil{!(isNil "BIS_fnc_init")}; _this spawn BIS_fnc_infoText; };

//Случайная стартовая позиция для ПВО
//0 = [[[pvo1,pvo2,pvo3],units pvo4],[pvopos,pvopos_1,pvopos_2,pvopos_3],2] execVM "scripts\bgr_RandomStart.sqf";

//Включаем отображение ников
0 = [] execVM "scripts\id.sqf";

/////////////////////////////////////////////////////////////
																																																;
/////////////////////////////////////////////////////////////
//То что ниже будет выполнено после окончания инициализации//
/////////////////////////////////////////////////////////////

//Пример ограничения на потери:
[] spawn {
	if (isServer) then {
		waitUntil { sleep 1;
			if ({alive _x} count bgr_playable_E < (count bgr_playable_E)*0.1) then {
				"Игроков красных осталось меньше 10%" call f_bgrEnd;
			};
			if ({alive _x} count bgr_playable_W < (count bgr_playable_W)*0.1) then {
				"Игроков синих осталось меньше 10%" call f_bgrEnd;
			};
			if ({alive _x} count bgr_playable_G < (count bgr_playable_G)*0.1) then {
				"Игроков зелёных осталось меньше 10%" call f_bgrEnd;
			};
			bgr_end
		};
	};
};

//Пример ограничения по времени
[] spawn {
	if (isServer) then {
		sleep 3600;
		"Прошёл чаc..." call f_bgrEnd;
	};
};

//Пример ограничения на выход из зоны игры
if (side player == west) then {
	0 = [player,["zoneB"],true,(getPosATL vehicle player),"false",0.5,0,false] call bgr_fnc_allowZone;
};
if (side player in [east,resistance,civilian]) then {
	0 = [player,["zoneR"],true,(getPosATL vehicle player),"false",0.5,0,false] call bgr_fnc_allowZone;
};